
【A-0】IchigoJamでシューティングゲームを作る 

 

 

●今回の目標 

 

IchigoJam で動くシューティングゲ

ームを作ります。 

自機を左右に動かして、ビームを

発射して、上空の UFO を打ち落

とします。 

 

 

 

 

 

●ゲームの画面を表示する 

 

まず、プログラムの初期設定をした後、画面にスコア（点数）を表示します。 

 

 

 

 

 

 

入力できたら、「RUN」で実行してみましょう。 

画面がクリアされて、左上に「0」が表示され

ます。 

 

 

 

 

プログラムの内容を説明します。 

 

10行：コメントで、プログラムのタイトルを入れています。 

先頭に「’」（アポストロフィ、キーボードでは Shift キーを押しながら「7」を押す）を付け

ると、その行はコメントとなり、何も実行されません。ですから、「’」のあとは好きな文字

を書くことができます。 

プログラムを後で見た時にわかりやすくするために、プログラムにいろいろコメントを

入れるといいでしょう。 

10 ‘*SHOOTING* 

20 CLS:CLV 

30 LC 0,0 

40 ?S 

コメントで、プログラムのタイトルを入れる 

スコアを表示 

カーソルを画面左上へ移動 

画面をクリア、変数をクリア 

20行：CLS命令で、画面をクリアします。 

その後の CLV（シーエルブイ）命令は、変数（へんすう）をクリアする命令です。 

変数とは、数字を入れる入れ物（箱）と思ってください。 

小学校の算数でやる「□」（四角）、中学校の数学でやる「x」と考えればいいです。 

今回は、ゲームの最初に、全ての変数をクリアしています。 

 

この行では、2 つの命令を「:」（コロン）で続けて書いています。「:」を使うと、複数の命

令を続けて書くことができます。 

 

30 行：LC（LOCATE、ローケート）命令は、画面に文字を表示する位置（カーソル位置）を設

定する命令です。 

 

LC 0 ,0 

 x座標 y座標 

 

x座標 画面の x座標（0～31）。 

y座標 画面の y座標（0～23）。 

 

IchigoJamの画面サイズは、横 32文字×縦 24

行になっています。 

横（x 方向）の座標は 0～31、縦（y 方向）の座

標は 0～23になっています。 

今回は「LC 0,0」として、画面左上の座標を指

定しています。 

 

40行：「?S」で、画面にスコア変数 Sの値を表示しています。 

（?は PRINT命令の略です） 

最初に CLV命令で変数をクリアしていて、Sの値も 0なので、「0」と表示されます。 

「?“S”」と S をダブルクォーテーションで囲むと、文字「S」を表示しますが、「?S」だと、

「変数 Sの値を表示する」という意味になります。 

 

LC命令の座標の数字を変えると、スコア「0」の表示位置が変わります。 

いろいろ変えて試してみましょう。 

●自機を表示する 

 

スコアと同じように、自機を表示してみましょう。 

以下のプログラムを追加します。 

 

 

 

 

 

 

 

 

 

プログラムを実行してみましょう。 

画面中央下に「A」が表示されます。 

 

追加したプログラムの内容を説明します。 

 

50 行：自機の横座標の変数 X、縦座標の

変数 Yに値を入れています。 

「X=15」は、「X と 15 が等しい」とい

う意味ではなく、「Xに 15を入れる」

（代入する）という意味です。 

 

60行：LC命令で、カーソルを(X,Y)の座標へ移動します。 

このように、座標を変数で指定することもできます。 

 

70行：?（PRINT）命令で、自機「A」を表示しています。 

ここでは「?“A”;」と最後に「;」（セミコロン）を付けています。 

PRINT 命令で文字を表示した後は、通常は自動的に改行してカーソルが次の行へ

移るのですが、「;」を付けると、改行せずに次の文字位置にカーソルがとどまります。 

（※改行するとあとで表示がおかしくなるので、こうしています） 

 

今の所は、「A」を表示した後にプログラムが終了して、続いて「OK」を表示してしまいます。 

これから続きのプログラムを作っていくので、今回は気にしないことにします。 

 

50行の変数 X,Y の値を変えれば、自機の表示位置が変わります。 

いろいろ変えて試してみましょう。 

  

10 ‘*SHOOTING* 

20 CLS:CLV 

30 LC 0,0 

40 ?S 

50 X=15:Y=22 

60 LC X,Y 

70 ?“A”; 自機を表示 

カーソルを座標(X,Y)へ移動 

自機の x座標を 15、y 座標を 22 にする 

●自機を動かす 

 

自機をカーソルキー（矢印キー）で左右に動かせるようにしてみましょう。 

以下のプログラムを追加します。 

 

 

 

 

 

 

 

 

 

 

 

 

 

プログラムを実行してみましょう。 

カーソルキーの左と右を押すと、自機が左

右に移動します。 

このままだとプログラムがいつまでも終わら

ないので、ESC キーを押して止めてくださ

い。 

 

 

 

 

 

追加したプログラムの内容を説明します。 

 

80行：ここからゲームの処理をするプログラムになるので、コメントを入れています。 

ループしてくり返すので、「GAMELOOP」としています。 

 

90～100行：自機を一度消します。 

  

…（前略）… 

60 LC X,Y 

70 ?“A”; 

80 ‘*GAMELOOP 

90 LC X,Y 

100 ?“ “; 

110 IF BTN(28)=1 THEN X=X-1 

120 IF BTN(29)=1 THEN X=X+1 

130 LC X,Y 

140 ?“A”; 

150 GOTO 80 

コメント 

左矢印キーが押されていたら 
自機の x 座標を 1 減らす 

自機を消す 

右矢印キーが押されていたら 
自機の x 座標を 1 増やす 自機を表示 

80 行へもどる 

110行：自機を左へ動かす処理をします。 

BTN（ボタン）関数を使って、左矢印キーが押されているかどうかを判断します。 

BTN関数の文法は以下のとおりです。 

 

BTN( 28) 
 キー指定 

 

キー指定 28（LEFT）…左矢印キー（←） 

29（RIGHT）…右矢印キー（→） 

30（UP）…上矢印キー（↑） 

31（DOWN）…下矢印キー（↓） 

32（SPACE）…スペースキー 

返り値 キーが押されている＝1 

キーが押されていない＝0 

 

指定したキーが押されていると BTN関数の値が 1、押されていないと 0になります。 

その BTN関数の値を、IF（イフ）命令で判断します。 

 

IF BTN(28)=1 THEN X=X-1 ELSE ～ 

 条件式  条件が成り立

つ時に実行 

 条件が成

り立たない

時に実行 

 

条件式 条件を判断する式。 

THEN（ゼン） 

の後 

条件が成り立つ時に実行するプログ

ラム。 

ELSE（エルス） 

の後 

条件が成り立たない時に実行するプ

ログラム。ELSE以下は省略可能。 

 

この行は、「もし BTN(28)が 1だったら（＝左矢印キーが押されていたら）、変数 Xを

1減らす」という意味になります。 

なお「X=X-1」は、「Xから 1を引いて、それを Xに入れる」という意味です。「X と X-

1が等しい」という意味ではないので注意してください。 

 

120行：110行と同じように、自機を右へ動かす（x座標を 1増やす）処理をします。 

 

130～140行：自機を新しい座標に表示します。 

 

150行：続けて自機を動かすため、GOTO（ゴートゥー）命令で前へ戻しています。 

 

GOTO 80 
 行番号 

指定した行番号へ実行を移します。 

このプログラムだと、以下の問題があります。 

➢ 自機が画面左はじへ行っても左矢印キーを押し続けると、自機は動かないが、次に

右へ動かそうとすると、左矢印キーを押していたのと同じ時間だけ右矢印キーを押

し続けないと、右へ動かない。 

➢ 右はじでも同様。 

これは、自機の横座標 X の範囲を考えずに増やしたり減らしたりしてしまい、X がマイナスの

値になってしまったりするからです。 

 

画面サイズを考えると、自機を動かす範囲

は、x 座標が 0～31 の間にしないといけませ

ん。 

 

 

 

 

自機が画面からはみ出さないように、プログ

ラムを改造します。 

 

 

 

 

 

 

 

 

 

 

 

IF命令の条件式に、「AND」（アンド）でつないで、２つの条件を入れます。 

 

 

この条件式は、「BTN(28)が 1 と等しい、かつ、X が 0 より大きい」という意味になります。 

両方の条件が成り立った時だけ、「THEN」以下の命令が実行されます。 

画面の左端に来ると、Xが 0になるので、条件が成り立たなくなり、それ以上左へ行きません。 

 

同じように、画面の右はじでもはみ出さないように、AND条件式を入れます。 

 

 

「SAVE 0」でプログラムを保存しておきましょう。 

…（前略）… 

80 ‘*GAMELOOP 

90 LC X,Y 

100 ?“ “; 

110 IF BTN(28)=1 AND X>0 THEN X=X-1 

120 IF BTN(29)=1 AND X<31 THEN X=X+1 

130 LC X,Y 

140 ?“A”; 

150 GOTO 80 

左矢印キーが押されている、 
かつ、自機の x座標が 0 より大きかったら、 
x座標を 1 減らす 

右矢印キーが押されている、 
かつ、自機の x座標が 31より小さかったら、 
x座標を 1 増やす 

110 IF BTN(28)=1 AND X>0 THEN X=X-1 

120 IF BTN(29)=1 AND X<31 THEN X=X+1 

●UFO を表示する 

 

自機の次は、敵の UFO を表示しましょう。 

まず、UFOの最初の設定をするプログラムを追加します。 

 

 

 

 

 

 

 

 

 

 

 

 

 

70行と 80行の間に行を追加するので、行番号を 72～76にしています。 

 

プログラムを実行してみましょう。 

UFOが画面に表示されます。 

プログラムを実行するたびに、UFO の

表示位置が変わります。 

 

 

 

 

 

 

 

次ページから、プログラムの内容を説明します。 

  

10 ‘*SHOOTING* 

20 CLS:CLV 

30 LC 0,0 

40 ?S 

50 X=15:Y=22 

60 LC X,Y 

70 ?“A”; 

72 U=RND(32):V=0 

74 LC U,V 

76 ?CHR$(241); 

80 ‘*GAMELOOP 

90 LC X,Y 

UFO の横座標 U を乱数で指定 

縦座標 V は 0 にする 

UFO を画面に表示 

72 行：UFO の横座標 U・縦座標 V を指定します。毎回同じ場所に出てくると面白くないの

で、Uは乱数（らんすう）で指定します。乱数とは、毎回違うでたらめな数字です。 

乱数を取り出すには、RND（ランダム）関数を使います。 

 

RND( 32) 
 乱数の最大値 

 

乱数の最大値 0～最大値-1の乱数が出てきます。 

 

「RND(32)」と指定すると、0～31の乱数が出てくるので、UFOの横座標 Uが画面の

左はじから右はじのどこかになります。 

縦座標 Vは、一番上（0）にします。 

 

74～76行：UFOを画面に表示します。ここでは UFOの文字を CHR$（シー・エッチ・アール・

ドル、キャラクター）関数で指定しています。 

 

CHR$( 241) 
 文字コード 

 

文字コード 0～255の数字で文字コードを指定 

 

今回は文字コード 241

番を指定して、UFO を

画面に表示しています。 

IchigoJam の文字コード

は、右のようになってい

ます。 

 

 

 

 

 

 

 

CHR$関数の文字コードを変えると、表示されるキャラクターが変わります。試してみましょう。 

自機の「A」を CHR$関数に変えて、自機のキャラクターをいろいろ変えるのも面白いでしょう。 

 

例  140 ?CHR$(240); 

  

UFO（241番） 

●UFO を動かす 

 

表示した UFOを動かしてみましょう。 

自機と UFO を同時に動かさないとゲームにならないので、自機を動かしているループのプ

ログラムから、UFOを動かすプログラムをサブルーチンとして呼び出す形にします。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

プログラムを実行してみましょう。UFOがランダムに動き回ります。 

それでは、プログラムの内容を説明します。 

 

145行：「GSB」（GOSUB、ゴーサブ）命令で、UFO を動かすサブルーチンを呼び出します。 

 

…（前略）… 

80 ‘*GAMELOOP 

90 LC X,Y 

100 ?“ “; 

110 IF BTN(28)=1 AND X>0 THEN X=X-1 

120 IF BTN(29)=1 AND X<31 THEN X=X+1 

130 LC X,Y 

140 ?“A”; 

145 GSB 160 

150 GOTO 80 

160 ‘*UFO 

170 LC U,V 

180 ?" "; 

190 U=U+RND(3)-1 

200 V=V+RND(3)-1 

210 LC U,V 

220 ?CHR$(241); 

230 RTN 

UFOを動かすサブルーチンを呼び出す 

UFO をいったん消す 

UFO の横座標・縦座標を乱数で変化 

UFO を表示 

メインプログラムへもどる 

UFO を動かすサブルーチン 



上田市マルチメディア情報センター 

 

【A-0】IchigoJamでシューティングゲームを作る／10 

 

メインプログラムからは「GSB」命令でサブルーチンへジャンプし、サブルーチンから

は「RTN」（RETURN、リターン）命令でもどります。もどった後は、GSB 命令の続きへ

プログラムの処理が移ります。 

サブルーチンに分けると、プログラムがすっきりしてわかりやすくなります。また、何度

も同じサブルーチンを呼び出して使うことができます。 

 

160行～：UFOを動かすサブルーチンです。 

 

170～180行：UFOをいったん消します。 

 

190～200行：UFOの横座標 U、縦座標

Vを乱数で変化させています。 

「RND(3)」で「0,1,2」のどれかの

数が出るので、「RND(3)-1」とす

ると「-1,0,1」のどれかになりま

す。これで、UFO の座標が縦横

に 1 ずつランダムに変化すること

になります。 

 

210～220行：UFOを表示します。 

 

230行：RTN命令でメインプログラムへも

どります。 

 

今のプログラムだと、UFO が画面の外へはみ出して消えてしまいます。 

自機の時と同じように、はみ出さないようにプログラムを改造します。 

 

 

 

 

 

 

 

 

プログラムを実行してみましょう。今度は UFOが画面からはみださなくなります。 

 

190行・200行の RND関数の最大値や引き算する値を変えると、UFOの動きが変わります。 

「SAVE 0」でプログラムを保存しておきましょう。 

190 U=U+RND(3)-1 

192 IF U<0 THEN U=0 

194 IF U>31 THEN U=31 

200 V=V+RND(3)-1 

202 IF V<0 THEN V=0 

204 IF V>22 THEN V=22 

210 LC U,V 

Uが 0より小さくなったら、0にもどす 

Uが 31 より大きくなったら、31 にもどす 

Vが 0 より小さくなったら、0にもどす 

Vが 22 より大きくなったら、22 にもどす 

●ビームを発射する 

 

自機からビームを発射してみましょう。 

まず、ゲームのメインループのプログラムで、「スペースキーが押されていたら、ビームのサブ

ルーチンを呼ぶ」ようにします。 

 

 

 

 

 

 

 

 

 

 

 

次に、プログラムの最後に、ビームのサブルーチンを追加します。 

 

 

 

 

 

 

 

 

 

 

プログラムを実行してみましょう。 

スペースキーを押すと、ビームが自

機から UFOの高さまで飛びます。 

 

 

 

 

 

 

 

 

 

80 ‘*GAMELOOP 

90 LC X,Y 

100 ?“ “; 

110 IF BTN(28)=1 AND X>0 THEN X=X-1 

120 IF BTN(29)=1 AND X<31 THEN X=X+1 

130 LC X,Y 

140 ?“A”; 

142 IF BTN(32)=1 THEN GSB 240 

145 GSB 160 

150 GOTO 80 

スペースキーが押されていたら、ビームのサブルーチンを呼ぶ 

240 '*BEAM 

250 B=X 

260 FOR C=Y-1 TO V STEP -1 

270 LC B,C 

280 ?"|"; 

290 LC B,C 

300 ?" "; 

310 NEXT 

320 RTN 

ビームの横座標 B を X にする（自機の横座標と同じ） 

ビームの縦座標 C を、自機の 
1 つ上から UFO の高さまで 
変化させてくり返し 

ビームを表示 

ビームを消去 

ここまでくり返し 

メインプログラムにもどる 

プログラムの内容を説明します。 

 

240行：ここからビームのサブルーチンが始まります。 

 

250行：ビームの横座標 Bを、自機の横座標Ｘにします。 

 

260 行：ビームを自機から UFO まで飛ばすため、FOR（フォー）命令と NEXT（ネクスト）命令

を使って、ビームの縦座標 Cを変化させてくり返します。 

 

FOR C=Y-1 TO V STEP -1 

 変数の 

初期値 

 変数の 

終値 

 変数の 

増分 

 

変数の初期値 ループ変数の最初の値 

変数の終値 ループ変数の最後の値 

変数の増分 ループ変数をどのくらい変化させるかの値 

STEP以下を省略すると１ずつ増やす 

 

 

ビームの動きを考えると、自機の１

つ上で発射されて、UFOまで上向

きに飛んでいきます。 

ですから、ビームの縦座標 C を、

自機の１つ上（Y-1）から、UFO の

縦座標（V）まで、1つずつ減らして

いけばいいことになります。 

 

この FOR 命令から、310 行の

NEXT 命令までのプログラムを、C

の値を Y-1 から V まで変化させて

くりかえします。 

 

270～280行：ビームを表示します。「|」の文字は、Shiftキーを押しながら「\」を押します。 

 

290～300行：ビームを消します。 

 

310行：くり返しの終わりを示す NEXT命令です。 

 

320行：RTN命令で、メインプログラムへもどります。 

このままだと、ビームの動きが速すぎてよく見えません。 

時間待ちの命令を入れて、動きを遅くしてみましょう。 

 

 

 

 

 

 

 

 

 

 

 

プログラムを実行してみましょう。 

今度はビームの動きが遅くなるので、UFOの高さまで飛ぶのがよくわかります。 

 

時間待ちをするために、WAIT（ウェイト）命令を使います。 

 

WAIT 2 
 待ち時間 

 

待ち時間 60分の 1秒単位で指定する。「60」で 1秒。 

 

WAITの待ち時間を変えると、ビームの移動速度が変わります。 

いろいろ変えて試してみましょう。 

 

「SAVE 0」でプログラムを保存しておきましょう。 

  

240 '*BEAM 

250 B=X 

260 FOR C=Y-1 TO V STEP -1 

270 LC B,C 

280 ?"|"; 

285 WAIT 2 

290 LC B,C 

300 ?" "; 

310 NEXT 

320 RTN 

時間待ちをする 

●ビームと UFO の当たり判定 

 

今のプログラムだと、ビームが UFOに当たっても何も起きません。 

ビームが UFOに当たったら、爆発するようにしましょう。 

 

 

 

 

 

 

 

 

 

 

 

プログラムを実行してみましょう。 

ビームが UFO に当たると、命中音が出て爆発

して、UFOが消えます。 

 

320行：ビームの横座標 Bと、UFOの横座標U

が違っていれば、ビームが当たってい

ないので、RTN 命令でメインプログラム

へもどります。 

「B!=U」は「B と Uが等しくない」という意

味の条件式です。 

 

330行：ビームが当たった時のプログラムが、ここから始まります。 

 

340行：BEEP（ビープ）命令で、命中音を出します。BEEP命令の文法は以下のとおりです。 

 

BEEP 30 ,30 
 音の高さ 音の長さ 

 

音の高さ 1～255で指定する。省略可能。 

音の長さ 60分の 1秒単位で指定する。「60」で 1秒。省略可能。 

 

数字を変えると、音の高さや長さが変わります。いろいろ変えて試してみましょう。 

 

350～360行：UFOの座標に、爆発「*」を表示します。 

310 NEXT 

320 IF B!=U THEN RTN 

330 '*HIT 

340 BEEP 

350 LC U,V 

360 ?"*"; 

370 WAIT 20 

380 LC U,V 

390 ?" "; 

400 RTN 

RETURN命令の行を書きかえ。もしビームが外れていたらもどる 

命中音を出す 

爆発の表示 

時間待ち 

表示を消す 

メインプログラムへもどる 

370行：爆発を表示した後に、しばらく時間待ちをします。 

 

380～390行：爆発表示を消します。 

 

400行：RTN命令で、メインプログラムへもどります。 

 

このままだと、UFOをうち落としても点数が入らないので、スコアを加算します。 

 

 

 

 

 

 

 

 

 

 

 

 

 

プログラムを実行してみましょう。 

UFO をうち落とすと、スコアが増えていくよ

うになります。 

 

362行：スコア変数 Sを 10点加算します。 

 

364～366行：画面左上にスコアを表示しま

す。 

 

 

362行の加算する数字を変えると、UFOをうった時の点数が変わります。 

いろいろ変えて試してみましょう。 

 

  

330 '*HIT 

340 BEEP 

350 LC U,V 

360 ?"*"; 

362 S=S+10 

364 LC 0,0 

366 ?S 

370 WAIT 20 

380 LC U,V 

390 ?" "; 

400 RTN 

スコア Sを 10 点加算 

スコアを画面に表示する 

UFOをうち落とした後、同じ場所にまた出てくると、次にうつのが簡単になってしまいます。 

UFOが上にもどって、違う場所に表示されるようにしましょう。 

 

 

 

 

 

 

 

 

 

 

 

 

 

最初の設定と同じように、UFOの位置を乱数で決めます。 

 

プログラムを実行してみましょう。 

UFOをうち落とすと、違う場所に出てくるので、次にうつのが難しくなります。 

 

 

「SAVE 0」でプログラムを保存しておきましょう。 

 

  

330 '*HIT 

340 BEEP 

350 LC U,V 

360 ?"*"; 

362 S=S+10 

364 LC 0,0 

366 ?S 

370 WAIT 20 

380 LC U,V 

390 ?" "; 

395 U=RND(32):V=0 

400 RTN 

UFO の位置を再設定する 

●UFO の侵略 

 

今のままだと、UFOが一方的にやられるだけなので、ゲームとして面白くありません。 

UFOが自機の段まで降りてきたら、ゲームオーバーになるように改造しましょう。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

「RND(10)」で乱数「0,1,2,…9」が出るので、それが 0 と等しい場合は（＝10分の 1の確率）、

UFOの縦座標 Vを 1増やします。これでだんだん下へ降りてくるようになります。 

 

そして UFOが一番下まで降りると、縦座標 Vが 22になるので、その場合はゲームオーバー

になるようにします。メインループのプログラムを改造します。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

160 ‘*UFO 

170 LC U,V 

180 ?" "; 

190 U=U+RND(3)-1 

192 IF U<0 THEN U=0 

194 IF U>31 THEN U=31 

200 V=V+RND(3)-1 

201 IF RND(10)=0 THEN V=V+1 

202 IF V<0 THEN V=0 

204 IF V>22 THEN V=22 

210 LC U,V 

220 ?CHR$(241); 

230 RTN 

10 分の 1 の確率で、 
UFO の縦座標 Vを 1 増やす 

80 ‘*GAMELOOP 

90 LC X,Y 

100 ?“ “; 

110 IF BTN(28)=1 AND X>0 THEN X=X-1 

120 IF BTN(29)=1 AND X<31 THEN X=X+1 

130 LC X,Y 

140 ?“A”; 

142 IF BTN(32)=1 THEN GSB 240 

145 GSB 160 

150 IF V<22 THEN GOTO 80 

152 BEEP 30,30 

154 LC 12,12 

156 ?"GAME OVER" 

158 CLK:END 

160 '*UFO 

Vが 22 より小さかったら、 
侵略されていないのでもどる 

ゲームオーバー音を出す 

画面中央にゲームオーバー表示 

キー入力クリア：プログラムを終了 

プログラムを実行してみましょう。 

UFO が自機の段まで来ると、ゲーム

オーバーになります。 

 

 

 

 

 

 

 

150行：これまで「GOTO 80」と無条件に前へもどっていた行を、IF命令を使って「Vが 22 よ

り小さかったらもどる」ようにします。 

 

152行：BEEP命令で、ゲームオーバーの音を出します。 

 

154～156行：画面の中央に「GAME OVER」と表示します。 

 

158 行：CLK（シーエルケー）命令でキー入力をクリアして、END（エンド）命令でプログラムを

終了します。 

 

 

201行の乱数の範囲を変えると、UFOが下に降りてくる速度が変わります。 

いろいろ変えて試してみましょう。 

 

 

 

 

 

「SAVE 0」でプログラムを保存しておきましょう。 

 

201 IF RND(10)=0 THEN V=V+1 


